Soundex is an algorithm for creating indices for words based on their pronunciation.



Task

The goal is for homophones to be encoded to the same representation so that they can be matched despite minor differences in spelling   (from the   soundex   Wikipedia article).

Caution

There is a major issue in many of the implementations concerning the separation of two consonants that have the same soundex code! According to the official Rules [[1]]. So check for instance if Ashcraft is coded to A-261.

 If a vowel (A, E, I, O, U) separates two consonants that have the same soundex code, the consonant to the right of the vowel is coded. Tymczak is coded as T-522 (T, 5 for the M, 2 for the C, Z ignored (see "Side-by-Side" rule above), 2 for the K). Since the vowel "A" separates the Z and K, the K is coded.
 If "H" or "W" separate two consonants that have the same soundex code, the consonant to the right of the vowel is not coded. Example: Ashcraft is coded A-261 (A, 2 for the S, C ignored, 6 for the R, 1 for the F). It is not coded A-226.


import strutils
 
const
  Wovel = 'W'   # Character code used to specify a wovel.
  Ignore = ' '  # Character code used to specify a character to ignore ('h', 'w' or non-letter).
 
 
proc code(ch: char): char =
  ## Return the soundex code for a character.
  case ch.toLowerAscii()
  of 'b', 'f', 'p', 'v': '1'
  of 'c', 'g', 'j', 'k', 'q', 's', 'x', 'z': '2'
  of 'd', 't': '3'
  of 'l': '4'
  of 'm', 'n': '5'
  of 'r': '6'
  of 'a', 'e', 'i', 'o', 'u', 'y': Wovel
  else: Ignore
 
proc soundex(str: string): string =
  ## Return the soundex for the given string.
 
  result.add str[0]   # Store the first letter.
 
  # Process characters.
  var prev = code(str[0])
  for i in 1..str.high:
    let curr = code(str[i])
    if curr != Ignore:
      if curr != Wovel and curr != prev:
        result.add curr
      prev = curr
 
  # Make sure the result has four characters.
  if result.len > 4:
    result.setLen(4)
  else:
    for _ in result.len..3:
      result.add '0'
 
 
for name in ["Robert", "Rupert", "Rubin", "Ashcraft", "Ashcroft", "Tymczak",
             "Pfister", "Honeyman", "Moses", "O'Mally", "O'Hara", "D day"]:
  echo name.align(8), " ", soundex(name)